ABSTRACT
Traditionally, tumor response and duration of treatment information have been displayed in separate plots in which the subjects can be sorted by different criteria. In such cases, the clinician must work harder to associate the subject across the plots. Recently there has been increased interest in combining this information in one visual. Displaying the data together, sorted by the tumor response with associated duration information, makes it easier for the clinician to understand this information. Three-dimensional waterfall plots, which have both pros and cons, have been proposed for such cases. This paper shows you how to build a 3-D plot using SAS® that shows both tumor response and duration of treatment. This paper also presents alternative 2-D visuals that were created using the SGPLOT procedure. These 2-D visuals enable easier decoding of the data, which enables you to display more information in the plot.

INTRODUCTION
The paper “3-D Waterfall Plots: A better graphical representation of tumor response in oncology” provides a compelling case for inclusion of more relevant data in plots for the analysis of tumor data. Classical waterfall plots are often complemented by separate swimmer plots and spider plots. The authors indicate that readers often need to consult different charts to get a complete understanding of the trial results. The authors propose to add the display of treatment duration into a 3-D waterfall plot to provide extra information that is useful to the reader.

Figure 1 shows a traditional waterfall plot. Figure 2 shows a 3-D waterfall plot where the tumor response is shown in the front vertical plane of the plot and the associated duration information for each subject is shown in the horizontal plane. The data for these plots is simulated and are for illustration purposes only.

Figure 1. Traditional Waterfall Plot
Figure 2. 3-D Waterfall Plot

Figure 1 shows a study’s tumor response sorted by increasing reduction in tumor size. The bars are colored by the treatment level and labeled by the type of lymphoma (DLBCL or FL) at the bottom of the bar. A band is displayed between +20% and -30% change in tumor size.

Figure 2 shows a 3-D plot where the tumor response is sorted by increasing reduction in tumor size and is displayed on the front vertical face of the plot. Again, the bars are colored by treatment level and the type of lymphoma is shown at the bottom of the bar. The duration of treatment for each subject is displayed on the horizontal plane of the plot.

3-D WATERFALL PLOT

The simulated data for the 3-D Waterfall plot is shown in Figure 3. The data is sorted by increasing reduction in tumor size. Data columns include the treatment, code, response, duration of treatment, and whether the subject was dropped from study. The code column is used to display the code value. The baseline column is not used in this plot.
Figure 3. Simulated Data for the 3-D Waterfall Plot

The plot shown in Figure 4 is created using the SAS macro %Waterfall_3D_Macro using the data shown in Figure 3. The macro is published in the Graphically Speaking SAS blog and requires the parameters in Table 1.

Table 1. 3-D Waterfall Plot Macro Parameters

<table>
<thead>
<tr>
<th>Parameter</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>Data</td>
<td>Data set</td>
</tr>
<tr>
<td>Duration</td>
<td>Column for Treatment Duration</td>
</tr>
<tr>
<td>Response</td>
<td>Column for Tumor Response</td>
</tr>
<tr>
<td>Dropped</td>
<td>Column for dropped subject</td>
</tr>
<tr>
<td>Group</td>
<td>Column for Treatment Group</td>
</tr>
<tr>
<td>Code</td>
<td>Column for Lymphoma code displayed at end of bar</td>
</tr>
</tbody>
</table>

The data is sorted by increasing reduction in tumor size (Response). The macro %Waterfall_3D_Macro lays out the provided data in 3-D space as follows to render the plot:
• The 3-D cube has a horizontal X axis, a Y-axis going into the page, and a Z-axis going up.
• All the bar data is coded as polygons in 3-D coordinate space.
• The tumor response bars are arranged on the front face (y=0) of the 3-D cube.
• Lymphoma type is displayed at the bottom of the response bars.
• The treatment duration bars are arranged on the horizontal plane at z=0.
• Markers are placed on the “duration” bars for dropped subjects.
• The 3-D data is transformed and projected into 2-D space.
• The 2-D polygonal data is displayed using the SGPLOT procedure.

![Tumor Response and Duration](image)

Figure 4. 3-D Waterfall Plot Generated with the SAS Macro %Waterfall_3D_Macro

While the 3-D plot shown in Figure 4 has a modern and slick appearance, there are some shortcomings of this visualization of the data:

• Some of the “duration” bars are hidden behind the “response” bars.
• The “dropped” markers on the duration bars are not clearly visible.
• There is a lot of unused space in this visual. Density of information pixels might be less than 20%.
• This is a custom bar layout and adding more data to this display is not easy.
• This data is really 2-D in nature with one independent variable (the sorted subject ID). All response, duration, and dropped information are dependent only the subject ID.

The only benefit of this arrangement is that the duration and response bars are touching in the middle, thus providing a connection between the data.
2-D WATERFALL PLOT WITH ADDITIONAL DATA

This paper proposes an alternate, simpler 2-D representation of the same data as shown in Figure 5. This representation has many advantages:

- The 2-D data is represented in a 2-D visual, without any “chart Junk.”
- The sorted response bars are shown in the lower cell, colored by treatment.
- The Lymphoma type is clearly labeled at the bottom of the bar.
- The duration for each subject is shown in the upper cell, clearly aligned with the bars in the lower cell.
- The duration value can be easily displayed at the top of each bar.
- The “dropped” markers are clearly visible.
- The vertical blue bands provide clear “connection” between the bars for each subject.
- Multiple legends are easier to place.
- The two Y axes for Duration and Response are clearly marked and readable.
- A band can be placed on the response bars indicating the +20 to -30 region.
- The “information density” is much higher, with better utilization of the space.
- This plot can be easily created as a 2-cell Lattice using Graph Template Language (GTL).

![Figure 5. 2-D Waterfall Plot with Duration Data](image)

Here is the template code for this example:

```plaintext
proc template;
define statgraph Waterfall_Plus;
begingraph / axislineextent=data;
entrytitle 'Tumor Response and Duration by Subject ID';
entryfootnote halign=left 'This graph uses simulated data for illustration only'/
textattrs=(size=7pt style=italic);
layout lattice / columndatarange=union rowweights=(0.45 0.55)
rowgutter=0;
columnaxes;
```
columnaxis / display=none discreteopts=(colorbands=odd
colorbandsattrs=(transparency=0.2));
endcolumnaxes;

/*--Define the upper cell with duration and dropped data--*/
layout overlay / yaxisopts=(griddisplay=on offsetmax=0.1
tickvalueattrs=(size=7) labelattrs=(size=9))
walldisplay=none;
barchartparm category=j response=duration / datalabel=duration
fillattrs=graphdata1 datalabelattrs=(size=5)
dataskin=pressed displaybaseline=auto;
scatterplot x=j y=dropped / markerattrs=(symbol=diamondfilled
size=9)
filledoutlinedmarkers=true
markerfillattrs=(color=gold)
markeroutlinattrs=(color=black)
name='d' legendlabel='Dropped';
discretelegend 'd' / location=inside valign=top halign=left
valueattrs=(size=7) border=false
autoitemsize=true;
endlayout;

/*--Define the lower cell for display of response data--*/
layout overlay / yaxisopts=(griddisplay=on tickvalueattrs=(size=7)
labelattrs=(size=9) offsetmax=0
linearopts=(tickvaluepriority=true))
walldisplay=none;
bandplot x=j limitupper=20 limitlower=-30 / extend=true
fillattrs=(color=gold transparency=0.75);
barchartparm category=j response=response / group=drug
groupdisplay=cluster datalabelattrs=(size=5)
dataskin=pressed name='a' datalabelfitpolicy=rotate;
textplot x=j y=codeloc text=code / rotate=90 position=left
textattrs=(size=6) contributeoffsets=(ymin);
discretelegend 'a' / location=inside valign=bottom halign=left
order=columnmajor down=3 opaque=true
valueattrs=(size=5) border=false;
endlayout;
endlayout;
endgraph;
end;
run;

/*--Render the graph using the defined template--*/
ods graphics / reset width=4in height=3in;
proc sgrender template=Waterfall_Plus data=tumorsorted dattrmap=attrmap;
format duration 3.0;
dattrvar drug="Resp";
run;

2-D WATERFALL PLOT WITH ADDITIONAL DATA

We can add more data displays to this plot. One example is adding the display of the tumor burden for each subject in the same plot. Since the same independent variable (subject ID) is used in that display, it is easy to extend this 2-D representation to accommodate more data as shown in Figure 6.
Figure 6 extends the previous plot as follows:

- A third cell is added below the “Response” cell in the plot.
- Tumor baseline values by subject ID are clearly displayed.
- The baseline values could be displayed as needles (as shown), scatter markers, or lines.
- The Y axis for baseline is clearly displayed.
- The +20 to -30 band is displayed with outlines only.

**Tumor Response and Duration by Subject Id**

![Graph showing tumor response and duration by subject ID](image)

*This graph uses simulated data for illustration only*

Figure 6. 2-D Waterfall Plot with Duration Data and Baseline Data

**WATERFALL PLOT WITH SWIMMER DATA**

In the previous examples, we have extended the tumor response data by adding the duration of treatment data for each subject. Additional columns can be added to display other related information as shown in Figure 6. Now, we will start with the data for creating a Swimmer Plot and extend it with tumor response information to create a combined Waterfall + Swimmer plot.

Figure 7 shows the data used to create a Swimmer plot, displaying the Tumor Response “story” for subjects in a study. The data includes one observation per subject showing the stage of the disease and one response interval. For subjects having multiple response durations, additional observations are included for the same subject “ID.” Disease stage and response status is included per episode.
Figure 7. Simulated Data for the Swimmer Plot with Added Columns for Tumor Response

The plot in Figure 8 shows a Swimmer Plot (Phillips 2014) created from the data in Figure 7 using the SGPLOT procedure.

Figure 8. Swimmer Plot of Subjects Receiving Treatment by Drug

The plot in Figure 8 shows the following details:

- One bar is displayed for each subject in the study colored by the disease stage.
- Continuing response is indicated by the arrow at the right end of the bar.
A discrete attributes map is used to color the disease stage with custom colors.
For each subject, the treatment episodes are displayed as overlaid lines with start and end markers.
The treatment episode is color coded by the status of the response.
Durable responders are indicated by the markers at the left end of the bars.
A legend of the disease stage is displayed at the bottom.
A legend of the response status and various markers is displayed in the plot at bottom right.

The SGPLOT procedure code for the Swimmer plot in Figure 8 is shown below:

```sas
proc sgplot data= swimmer dattrmap=attrmap nocycleattrs noborder;
    legenditem type=marker name='ResStart' /
        markerattrs=(symbol=trianglefilled color=darkgray size=9)
        label='Response start';
    legenditem type=marker name='ResEnd' / label='Response end'
        markerattrs=(symbol=circlefilled color=darkgray size=9);
    legenditem type=marker name='RightArrow' / label='Continued response'
        markerattrs=(symbol=trianglerightfilled color=darkgray size=12);
    highlow y=item low=low high=high / highcap=highcap type=bar
        group=stage fill nocycleline dataSkin=gloss lineattrs=(color=black)
        name='stage' barwidth=1 nomissinggroup transparency=0.3
        attrid=stage;
    highlow y=item low=startline high=endline / group=status
        lineattrs=(thickness=2 pattern=solid) name='status'
        nomissinggroup attrid=status;
    scatter y=item x=durable / markerattrs=(symbol=squarefilled size=6
        color=black) name='Durable' legendlabel='Durable responder';
    scatter y=item x=start / markerattrs=(symbol=trianglefilled size=8)
        group=status attrid=status;
    scatter y=item x=end / markerattrs=(symbol=circlefilled size=8)
        group=status attrid=status;
    xaxis display=(nolabel) values=(0 to 20 by 1) valueshint grid;
    yaxis reverse display=(noticks novalues noline)
        label='Subjects Received Study Drug' min=1;
    keylegend 'stage' / title='Disease Stage';
    keylegend 'status' 'Durable' 'ResStart' 'ResEnd' 'RightArrow' /
        noborder location=inside position=bottomright across=1
        linelength=20;
run;
```

Since we added tumor response data for each subject in the data shown in Figure 7, we can create a traditional waterfall plot as shown in Figure 9.
The SGPLOT procedure code for the waterfall plot in Figure 9 is shown below:

```plaintext
title 'Change in Tumor Size';
proc sgplot data= swimmer_sort_2 noborder;
vbarparm category=id response=response / dataskin=pressed;
refline 20 -30 / lineattrs=(pattern=shortdash);
xaxis display=none;
yaxis label='Tumor Response';
run;
```

We can create a combined Waterfall + Swimmer plot as shown in Figure 10 from the data in Figure 7. Additional information can be added to this plot in cells above or below. The details of this plot are as follows:

- The tumor response is shown in the lower cell, very similar to the plot in Figure 9.
- The subject response history is shown in the upper cell, aligned with the bar for tumor response.
- The plot in the upper cell is essentially a rotated version of the plot in Figure 8.
- The subject response shows a vertical bar for each subject for the duration of the response by stage.
- Continued response is indicated by the arrow head.
- Individual treatment episodes are displayed by the line with start and end markers by status.
- Durable responders are indicated by the markers at the bottom of each bar.
- A legend for the disease stage is shown at the top.
- A legend for the status and other markers is shown inside the upper cell.
- Vertical alternate blue bands help align the related bars.
Figure 10. Combined Waterfall and Swimmer Plot

The GTL code for the combined plot in Figure 10 is shown below:

```gtl
proc template;
define statgraph Swimmer_With_Response;
begingraph / axislineextent=data;
  entrytitle 'Tumor Response with Duration by Stage and Month';
  entryfootnote halign=left
    'This graph uses simulated data for illustration' /
    textattrs=(size=7pt style=italic);
  legenditem type=marker name='ResStart' / label='Response start'
    markerattrs=(symbol=squarefilled color=darkgray size=7);
  legenditem type=marker name='ResEnd' / label='Response end'
    markerattrs=(symbol=circlefilled color=darkgray size=7);
  legenditem type=marker name='RightArrow' / label='Continued response'
    markerattrs=(symbol=trianglefilled color=darkgray size=12);

  layout lattice / columndatarange=union rowweights=(0.6 0.4)
    rowgutter=0;
columnaxes;
columnaxis / display=none type=discrete
discreteopts=(colorbands=odd
  colorbandsattrs=(transparency=0.1));
```

This graph uses simulated data for illustration only
With some modification of this template, this same plot can be arranged horizontally (Figure 11). The changes primarily involve the following:

- Set the number of LATTICE columns to two (columns=2).
- Change ROWWEIGHTS to COLUMNWEIGHTS.
- Change COLUMNAXES to ROWAXES.
• Change YAXISOPTS to XAXISOPTS.
• Change variable assignments to plot roles from X to Y.

Tumor Response with Duration by Stage and Month

Disease Stage:  ❖ Stage 1  ❖ Stage 2  ❖ Stage 3  ❖ Stage 4

Legend:
- Complete response
- Partial response
- Durable response
- Response start
- Response end
- Continued response

This graph uses simulated data for illustration only

Figure 11. Combined Waterfall and Swimmer Plot Shown Horizontally

The template code for Figure 11 is shown below:

```sas
proc template;
   define statgraph Swimmer_Plus_Horizontal;
   begingraph / axislineextent=data;
      entrytitle 'Tumor Response with Duration by Stage and Month';
      entryfootnote halign=left 'This graph uses simulated data for illustration only' /
         textattrs=(size=7pt style=italic);
      legenditem type=marker name='ResStart' /
         markerattrs=(symbol=squarefilled color=darkgray size=7)
         label='Response start';
      legenditem type=marker name='ResEnd' /
         markerattrs=(symbol=circlefilled color=darkgray size=7)
         label='Response end';
      legenditem type=marker name='RightArrow' /
         markerattrs=(symbol=trianglerightfilled color=darkgray size=9)
         label='Continued response';
      layout lattice / rowdatarange=union columnweights=(0.25 0.75)
         columngutter=0 columns=2;
         rowaxes;
         rowaxis / display=none type=discrete reverse=true
            discreteopts=(colorbands=odd
                          colorbandsattrs=(transparency=0.1));
      endrowaxes;

      layout overlay / xaxisopts=(griddisplay=on
tickvalueattrs=(size=6)
labelattrs=(size=7)
walldisplay=none
linearopts=(tickvaluepriority=true)
label='Change from Baseline (%)');
      bandplot y=id limitupper=20 limitlower=-30 /
```
CONCLUSION

There is an increasing interest in plotting tumor response and duration of treatment data together in one plot so that it is easier for the reader of the plot to understand the information. To view duration of treatment together with tumor response, 3-D visuals have been suggested. These 3-D visuals of 2-D data are difficult to read and make inefficient use of the space available in the plot. Adding more data makes these plots more cluttered.

This paper presents alternative 2-D visuals that present the same data in a clean and understandable visual with better usage of the space available. Comparisons are easier as
we use linear comparisons from common baselines. Information can be more easily aligned, and additional information can be displayed without clutter.

Swimmer plots present more information about a subject’s treatment history. This data can be combined with tumor response data and plotted together as a combined Waterfall + Swimmer plot for better understanding of the data.

REFERENCES


RECOMMENDED READING
- 3D WaterFall Macro: https://blogs.sas.com/content/graphicspeaking. Sanjay Matange, SAS Institute Inc.

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:

Sanjay Matange
(919) 210-0498
sanjaymatange@gmail.com
https://make-a-graph.com/

Dan Heath
SAS Institute, Inc.
(919) 677-8000
Dan.Heath@sas.com
https://blogs.sas.com/content/graphicspeaking

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.